
1. [bookmark: _Toc122434485][bookmark: _Toc109997190]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D22wk23 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc109997191]Corrections required


1. [bookmark: _Toc30685521][bookmark: _Toc109997192]Change 1
	Function name
	f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8()

	Reason for change
	Following to 38.523-3 cl 7.3.5.3.2, the time T “T is set to 300 ms in advance of the handover”. Step 2 (sending of RRCReconfiguration) occurs at time T, setp 1 (Getting PDCP counts from SS) occurs “before T”).

But, in the  implementation of this function, the time T is calculated from the current system time before calling f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(), and so the implementation of step 1 included in the time. The retrieval of PDCP soiunts fomr the SS is actually performed twice (once through f_NR_GetActiveDRBs() and then again through f_NR_GetPdcpCountInfoListIntraNR_HO(). This overhead eats into the 300ms and can cause a UE to fail this at this step.

	Summary of change
	Calculate the next send occasion and apply the 300ms advance just before step 2, in line with the procedure description.

Note : The parameter p_TimingInfo is now not used but is left in place to avoid changes in the calling functions. If this parameter is deleted it will require further changes in f_NR5GC_508RRC_IntraNR_HO_IntraCell_Common() and fl_TC_7_1_3_4_2_NR5GC_TestBody(). 

Note : The calculated time is now returned from the function to satisfy f_NR_RLC_ReconfigurationHO_Steps1_8(), see change 2

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	



Before Change
	function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_NR_CellId,
                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,
                                                       boolean                                p_RBConfig_KeyChange := true,
                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,
                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,
                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,
                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                       template (omit) MeasConfig             p_MeasConfig := omit,
                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,
                                                       boolean p_SetGetOfPDCPCount := true
                                                       ) runs on NR_BASE_PTC
  { //Added p_SetGetOfPDCPCount to be set to false in PDCP tc with TM mode
    var SubFrameTiming_Type v_Timing;
    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_NR_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    
    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).
    // Not needed in PDCP tc with PDCP configured in TM mode
    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_NR_CellId, p_NR_RadioBearerList, p_RBConfig_KeyChange); //@sic R5s220123 sic@
    // @sic R5s191200 R5s201150 sic@

   /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.
    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */
    if (f_TimingInfo_IsNow(p_TimingInfo)) {
      v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);
    } else {
      v_Timing := valueof(p_TimingInfo.SubFrame);
    }
    
    //Step 2 At T: Send RRCReconfiguration.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));
    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration
       => RLC ACK is not taken into consideration */
    
    //Step 3 At T + 5ms: Release SRBs and DRBs.
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Release(p_NR_CellId, cs_TimingInfo_NR(v_Timing), v_DrbIdList, tsc_NoCnfReq, tsc_FollowOnFlag);   //@sic R5-206369 sic@

    // Step 4 At T + 5ms: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE: Since the RACH procedure may require a new C-RNTI to be used it cannot be configured before sending out the RRCReconfiguration.
    // NOTE 2: The FollowOnFlag is set to true in the ASPs reconfiguring the RadioBearerList (step 3) and C-RNTI (step 4).
    f_NR_SS_C_RNTI_Config(p_NR_CellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing), tsc_NoCnfReq, tsc_FollowOnFlag); //@sic R5s190747 change 5.3 R5-206369 sic@
    f_NR_SS_RachProcedureConfigHO(p_NR_CellId, p_RachProcedureConfig, cs_TimingInfo_NR(v_Timing), p_RACH_ConfigDedicated);
    f_NR_CellInfo_SetRNTI(p_NR_CellId, p_RNTI_Value);
          
    // Step 5 At T + 5ms: (Re-)configure measurement gap if provided in the RRCReconfiguration message.
    // @sic R5-197222 sic@
    if (isvalue(p_MeasConfig.measGapConfig)) {
        f_NR_SS_MeasGapCtrlConfig(p_NR_CellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig), cs_TimingInfo_NR(v_Timing));
    }

    // Step 6 At T + 10ms: (Re-) configure SRBs and DRBs.
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Config(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_NR_RadioBearerList);

    // Step 7 At T + 10ms:  Restore the PDCP count for AM DRBs and SRBs (if applied).
    //NOTE 2:   For AM DRBs the PDCP count is maintained while for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the  RRCReconfiguration message content
    // Not needed in PDCP tc with PDCP configured in TM mode
    if (p_SetGetOfPDCPCount){
        f_NR_SS_PdcpCount_Set(p_NR_CellId, v_NR_PdcpCountInfoList, cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
    }
   // Step 8 At T + 10ms: Re-establish security, disable TA transmission.
   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.
   if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s191200 R5s200309 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_NR_CellId, p_NCC, cs_TimingInfo_NR(v_Timing),-, v_NR_PdcpCountInfoList); //@sic R5s200754 R5s220123 sic@
   } else {
      v_RRC_SecurityParams := f_NR_Security_Get();
      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false, -, v_NR_PdcpCountInfoList); //@sic R5s220123 sic@
      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering,
                                            cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
   }
  
   // Disable TA transmission
   f_NR_ULGrantConfiguration_Common(p_NR_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);

  }



After Change
	  function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_NR_CellId,
                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,
                                                       boolean                                p_RBConfig_KeyChange := true,
                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,
                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,
                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,

                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                       template (omit) MeasConfig             p_MeasConfig := omit,
                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,
                                                       boolean p_SetGetOfPDCPCount := true
                                                       ) runs on NR_BASE_PTC return SubFrameTiming_Type 
  { //Added p_SetGetOfPDCPCount to be set to false in PDCP tc with TM mode
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_TimingRRC;
    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_NR_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    
    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).
    // Not needed in PDCP tc with PDCP configured in TM mode
    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_NR_CellId, p_NR_RadioBearerList, p_RBConfig_KeyChange); //@sic R5s220123 sic@
    // @sic R5s191200 R5s201150 sic@

   /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.
    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */
    v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);
   
    v_TimingRRC := v_Timing; 
    
    //Step 2 At T: Send RRCReconfiguration.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));
    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration
       => RLC ACK is not taken into consideration */
    
    //Step 3 At T + 5ms: Release SRBs and DRBs.
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Release(p_NR_CellId, cs_TimingInfo_NR(v_Timing), v_DrbIdList, tsc_NoCnfReq, tsc_FollowOnFlag);   //@sic R5-206369 sic@

    // Step 4 At T + 5ms: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE: Since the RACH procedure may require a new C-RNTI to be used it cannot be configured before sending out the RRCReconfiguration.
    // NOTE 2: The FollowOnFlag is set to true in the ASPs reconfiguring the RadioBearerList (step 3) and C-RNTI (step 4).
    f_NR_SS_C_RNTI_Config(p_NR_CellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing), tsc_NoCnfReq, tsc_FollowOnFlag); //@sic R5s190747 change 5.3 R5-206369 sic@
    f_NR_SS_RachProcedureConfigHO(p_NR_CellId, p_RachProcedureConfig, cs_TimingInfo_NR(v_Timing), p_RACH_ConfigDedicated);
    f_NR_CellInfo_SetRNTI(p_NR_CellId, p_RNTI_Value);
          
    // Step 5 At T + 5ms: (Re-)configure measurement gap if provided in the RRCReconfiguration message.
    // @sic R5-197222 sic@
    if (isvalue(p_MeasConfig.measGapConfig)) {
        f_NR_SS_MeasGapCtrlConfig(p_NR_CellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig), cs_TimingInfo_NR(v_Timing));
    }

    // Step 6 At T + 10ms: (Re-) configure SRBs and DRBs.
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Config(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_NR_RadioBearerList);

    // Step 7 At T + 10ms:  Restore the PDCP count for AM DRBs and SRBs (if applied).
    //NOTE 2:   For AM DRBs the PDCP count is maintained while for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the  RRCReconfiguration message content
    // Not needed in PDCP tc with PDCP configured in TM mode
    if (p_SetGetOfPDCPCount){
        f_NR_SS_PdcpCount_Set(p_NR_CellId, v_NR_PdcpCountInfoList, cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
    }
   // Step 8 At T + 10ms: Re-establish security, disable TA transmission.
   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.
   if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s191200 R5s200309 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_NR_CellId, p_NCC, cs_TimingInfo_NR(v_Timing),-, v_NR_PdcpCountInfoList); //@sic R5s200754 R5s220123 sic@
   } else {
      v_RRC_SecurityParams := f_NR_Security_Get();
      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false, -, v_NR_PdcpCountInfoList); //@sic R5s220123 sic@
      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering,
                                            cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
   }
  
   // Disable TA transmission
   f_NR_ULGrantConfiguration_Common(p_NR_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);
   
   return v_TimingRRC;
  }



1. [bookmark: _Toc109997193][bookmark: _Toc30685522]Change 2
	Function name
	f_NR_RLC_ReconfigurationHO_Steps1_8()

	Reason for change
	As change 1

	Summary of change
	Return the timing form f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(), see change 1

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	



Before Change
	  function f_NR_RLC_ReconfigurationHO_Steps1_8(NR_CellId_Type p_NR_CellId,
                                               DRB_Identity p_DRBId,
                                               template (value) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList,
                                               template (value) PDCP_Config p_PDCP_Config,
                                               template (omit) NR_RadioBearer_Type p_SS_Drb_Config := omit,
                                               boolean p_OmitRecoverPDCP := false,
                                               template (omit) CellGroupConfig p_CellGroupConfig := omit
                                              ) runs on NR5GC_PTC return SubFrameTiming_Type
  { //@sic R5-194296 R5-204512 sic@
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RadioBearer_Type v_SS_Drb_Config;
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_NR_CellId);
    var SubFrameTiming_Type v_Timing;

    if (ispresent(p_CellGroupConfig)) { // @sic R5s191200 sic@
      v_CellGroupConfig := p_CellGroupConfig;
    } else {
      v_CellGroupConfig := f_NR_GetCellGroupConfigHO (p_NR_CellId, p_RLC_BearerConfigList); //tsc_C_RNTI_Value3
    }

    v_RadioBearerConfig := f_NR5GC_GetExisting_RadioBearerConfig(false, p_PDCP_Config); // RBConfig_NoKeyChange
    if (p_OmitRecoverPDCP) { // @sic R5s200180 sic@
        v_RadioBearerConfig.drb_ToAddModList[0].recoverPDCP := omit;
    }
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));
    v_RRCReconfiguration := cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext);
    if (isvalue(p_SS_Drb_Config)) { // @sic R5s190905_r1 sic@
      v_SS_Drb_Config := p_SS_Drb_Config;
    } else {
      v_SS_Drb_Config := cs_NR_SS_RadioBearer_AM_Test(p_DRBId, -); // size18;
    }
    v_RachProcedureConfig := cs_NR_RachProcedureConfig_CRNTI_HO(v_NR_PhysicalParameters, tsc_C_RNTI_Value3);
    v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);
    
    f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(p_NR_CellId, v_RRCReconfiguration, false, {v_SS_Drb_Config}, v_RachProcedureConfig, cs_TimingInfo_NR(v_Timing), tsc_C_RNTI_Value3, -, -, -);

    return v_Timing;
  }



After Change
	  function f_NR_RLC_ReconfigurationHO_Steps1_8(NR_CellId_Type p_NR_CellId,
                                               DRB_Identity p_DRBId,
                                               template (value) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList,
                                               template (value) PDCP_Config p_PDCP_Config,
                                               template (omit) NR_RadioBearer_Type p_SS_Drb_Config := omit,
                                               boolean p_OmitRecoverPDCP := false,
                                               template (omit) CellGroupConfig p_CellGroupConfig := omit
                                              ) runs on NR5GC_PTC return SubFrameTiming_Type
  { //@sic R5-194296 R5-204512 sic@
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RadioBearer_Type v_SS_Drb_Config;
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_NR_CellId);
    var SubFrameTiming_Type v_Timing;

    if (ispresent(p_CellGroupConfig)) { // @sic R5s191200 sic@
      v_CellGroupConfig := p_CellGroupConfig;
    } else {
      v_CellGroupConfig := f_NR_GetCellGroupConfigHO (p_NR_CellId, p_RLC_BearerConfigList); //tsc_C_RNTI_Value3
    }

    v_RadioBearerConfig := f_NR5GC_GetExisting_RadioBearerConfig(false, p_PDCP_Config); // RBConfig_NoKeyChange
    if (p_OmitRecoverPDCP) { // @sic R5s200180 sic@
        v_RadioBearerConfig.drb_ToAddModList[0].recoverPDCP := omit;
    }
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));
    v_RRCReconfiguration := cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext);
    if (isvalue(p_SS_Drb_Config)) { // @sic R5s190905_r1 sic@
      v_SS_Drb_Config := p_SS_Drb_Config;
    } else {
      v_SS_Drb_Config := cs_NR_SS_RadioBearer_AM_Test(p_DRBId, -); // size18;
    }
    v_RachProcedureConfig := cs_NR_RachProcedureConfig_CRNTI_HO(v_NR_PhysicalParameters, tsc_C_RNTI_Value3);
    //REMOVED v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);

    v_Timing := f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(p_NR_CellId, v_RRCReconfiguration, false, {v_SS_Drb_Config}, v_RachProcedureConfig, tsc_C_RNTI_Value3, -, -, -); 

    return v_Timing;
  }



