
1. [bookmark: _Toc122434485][bookmark: _Toc52895930]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D22wk23 related to the title of this CR.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54186268]Corrections required
2.1 Change 1
	Function name
	f_TC_11_2_7_IMS2

	Reason for change
	The functions f_IMS_EmergencyRegistration & f_IMS_EmergencyCallSetup (running in parallel) makes use of the same altstep (a_IMS_ReceiveRequest) to receive Subsequent REGISTER and INVITE respectively. Hence leading the TTCN to receive either one but not both. This may cause Runtime error when UE performs optional emergency registration during Step14-14L.

	Summary of change
	Both REGISTER & INVITE are received by same receive event and later checked for the content to perform either emergency registration or emergency call accordingly.  

	TTCN module
	IMS_EmergencyCall_IMS.ttcn

	MCC160 Comment
	



Before change:
function f_TC_11_2_7_IMS2() runs on IMS_PTC
  {
    var INVITE_Request v_InviteRequest;

    var default v_RegistrationHandler := null;

    f_IMS_PTC_Init(PDN_2);
    
    // Emergency registration: 34.229-1 Annex C.20
    f_IMS_EmergencyRegistration();
    
    // Emergency call setup: 34.229-1 Annex C.22
    v_InviteRequest := f_IMS_EmergencyCallSetup();

    //Release Emergency Call MO
    f_IMS_EmergencyCallRelease(v_InviteRequest);
    
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    // 2nd emergency registration may be optional: default handler to deal with C.20 if happening
    v_RegistrationHandler := activate(a_IMS_RcvAndHandleRegister(car_IMS_Register_Request(cr_REGISTER_Request())));  //@sic R5-191230 sic@

    // Emergency call setup: 34.229-1 Annex C.22
    v_InviteRequest := f_IMS_EmergencyCallSetup();

    // deactivate default handler to deal with C.20 if happening
    deactivate(v_RegistrationHandler);
    
    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
     
    //IMS call to be released.
    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    //Release Emergency Call MO
    f_IMS_EmergencyCallRelease(v_InviteRequest);
    
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
  }

  /*
   * @desc      Catches emergency registration according to 34.229-1 Annex C.20
   * @param     p_IMS_DATA_REQ
   * @status    APPROVED (LTE)
   */
  altstep a_IMS_RcvAndHandleRegister(template (present) IMS_DATA_REQ p_IMS_DATA_REQ) runs on IMS_PTC
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    
    [] IMS_Server.receive(p_IMS_DATA_REQ) -> value v_IMS_DATA_REQ
      {
        //Release Register info
        f_IMS_GlobalRelease();
        // Emergency registration: 34.229-1 Annex C.20
        f_IMS_EmergencyRegistration(v_IMS_DATA_REQ);
      }
  }

After change:

function f_TC_11_2_7_IMS2() runs on IMS_PTC
  {
    var INVITE_Request v_InviteRequest;
	var IMS_DATA_REQ v_IMS_DATA_REQ;
	var default v_RetransmissionHandler := null;

    f_IMS_PTC_Init(PDN_2);
    
    // Emergency registration: 34.229-1 Annex C.20
    f_IMS_EmergencyRegistration();
    
    // Emergency call setup: 34.229-1 Annex C.22
    v_InviteRequest := f_IMS_EmergencyCallSetup();

    //Release Emergency Call MO
    f_IMS_EmergencyCallRelease(v_InviteRequest);
    
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    // Receive 2nd optional emergency registration / INVITE for emergency call
	IMS_Server.receive(cr_IMS_DATA_REQ_Any) -> value v_IMS_DATA_REQ;
	
	// Handle Retransmissions of REGISTER / INVITE that may be received by SS, due to overhead of following few functions post reception of above
	v_RetransmissionHandler := activate(a_IMS_RetransmissionHandler(v_IMS_DATA_REQ)); 
	
	if (ischosen(v_IMS_DATA_REQ.Request.Register)) {
	  //Release Register info
	  f_IMS_GlobalRelease();
	  
	  // Emergency registration: 34.229-1 Annex C.20
	  f_IMS_EmergencyRegistration(v_IMS_DATA_REQ);
	  
	  // Emergency call setup: 34.229-1 Annex C.22
	  v_InviteRequest := f_IMS_EmergencyCallSetup();
	}
	else if (ischosen(v_IMS_DATA_REQ.Request.Invite)) {
	  // Initialise global variables for SIP dialog
	  f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);
	  
	  // Optional
	  f_IMS_A_2_1_Invite_CommonChecks(valueof(v_IMS_DATA_REQ), A_2_1_A7);
	  
	  // Emergency call setup: 34.229-1 Annex C.22
	  v_InviteRequest := f_IMS_EmergencyCallSetup(-,-,-,v_IMS_DATA_REQ);
	}
	
    deactivate(v_RetransmissionHandler);

    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
     
    //IMS call to be released.
    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    //Release Emergency Call MO
    f_IMS_EmergencyCallRelease(v_InviteRequest);
    
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
  }
	
  // REMOVED altstep a_IMS_RcvAndHandleRegister
2.2 Change 2
	Function name
	f_IMS_EmergencyCallSetup

	Reason for change
	Required to fulfil Change 1

	Summary of change
	Function is made to accept IMS_DATA_REQ as parameter and assign it to the variable v_IMS_DATA_REQ, if the parameter has a value.

	TTCN module
	IMS_Procedures_CallControl4G5G.ttcn

	MCC160 Comment
	



Before change:
function f_IMS_EmergencyCallSetup(INVITE_A_2_1_Context_Type p_Context := A_2_1_A7,
                                    boolean p_GeolocationInfoAvailable := false,
                                    template SipUrl p_OtherCalleeUri := omit) runs on IMS_PTC return INVITE_Request
  { /* p_Context   ..   A_2_1_A6 (INVITE for creating an emergency session in case of no registration) or
     *                  A_2_1_A7 (INVITE for creating an emergency session within an emergency registration) */
    /* @sic R5s130988 change 6/7: sic@
     * in case of A_2_1_A6 there has been no registration beforehand and therefore we don't know whether the UE uses IPv4 or IPv6
     * => we cannot setup any address information before we have received the INVITE
     * => as the SDP message at step 1 contains address information we cannot build up the template in advance but need to match after the INVITE has been received */
    /* @sic R5-140920: additional parameter p_OtherCalleeUri sic@ */
    /* @sic R5s150204: p_OtherCalleeUri charstring -> SipUrl sic@ */
    /* @sic R5s210875: split function into step 1 and the rest sic@ */
    var IMS_DATA_REQ v_IMS_DATA_REQ;

    // Step 1. Receive INVITE for an Emergency call
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_Step1(p_Context, p_GeolocationInfoAvailable, p_OtherCalleeUri);
	
    return f_IMS_EmergencyCallSetup_Step2ToEnd(v_IMS_DATA_REQ, p_GeolocationInfoAvailable);
  }

After change:

function f_IMS_EmergencyCallSetup(INVITE_A_2_1_Context_Type p_Context := A_2_1_A7,
                                    boolean p_GeolocationInfoAvailable := false,
                                    template SipUrl p_OtherCalleeUri := omit,
									template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit) runs on IMS_PTC return INVITE_Request
  { /* p_Context   ..   A_2_1_A6 (INVITE for creating an emergency session in case of no registration) or
     *                  A_2_1_A7 (INVITE for creating an emergency session within an emergency registration) */
    /* @sic R5s130988 change 6/7: sic@
     * in case of A_2_1_A6 there has been no registration beforehand and therefore we don't know whether the UE uses IPv4 or IPv6
     * => we cannot setup any address information before we have received the INVITE
     * => as the SDP message at step 1 contains address information we cannot build up the template in advance but need to match after the INVITE has been received */
    /* @sic R5-140920: additional parameter p_OtherCalleeUri sic@ */
    /* @sic R5s150204: p_OtherCalleeUri charstring -> SipUrl sic@ */
    /* @sic R5s210875: split function into step 1 and the rest sic@ */
    var IMS_DATA_REQ v_IMS_DATA_REQ;
	
	if (not isvalue(p_IMS_DATA_REQ)) {
      // Step 1. Receive INVITE for an Emergency call
      v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_Step1(p_Context, p_GeolocationInfoAvailable, p_OtherCalleeUri);
    } else {
      v_IMS_DATA_REQ := valueof(p_IMS_DATA_REQ);
    }
	
    return f_IMS_EmergencyCallSetup_Step2ToEnd(v_IMS_DATA_REQ, p_GeolocationInfoAvailable);
  }

